

Выпадающие списки

Выпадающие (раскрывающиеся) списки на листах Microsoft Excel – крайне полезный инструмент и используются очень часто. Во-первых, это удобно и позволяет не вводить повторяющиеся элементы списков с клавиатуры. Во-вторых, это исключает "человеческий фактор" и, во многих случаях, уберезет от ошибок и опечаток при вводе данных.

В этой главе вы узнаете:

- 4 способа создать **выпадающий список** на листе Excel.
- Как создать **связанные друг с другом списки** (содержимое второго зависит от выбора в первом).
- Как создать выпадающий **список с картинками**.
- Как создать выпадающие **списки с автоматическим добавлением или удалением элементов**.
- Как сделать выпадающий **список с данными из другого файла**.



Как создать выпадающий список

Способ 1. Быстрый

Один щелчок правой кнопкой мыши по пустой ячейке под столбцом с данными, команда контекстного меню **Выбрать из раскрывающегося списка (Choose from drop-down list)** или нажать сочетание клавиш **Alt+стрелка вниз**. Появится отсортированный список уникальных ранее введенных значений:

	A	B	C	D	E
1	Наименование	Месяц	Склад	Продано	
2	Апельсины	Январь	#001	25	
3	Груши	Февраль	#001	311	
4	Киви	Апрель	#002	623	
5	Яблоки	Февраль	#002	207	
6	Апельсины	Март	#001	415	
7					
	<div style="border: 1px solid black; padding: 2px;"> Апельсины Груши Киви Яблоки </div>				
11					
12					

Способ не работает, если ячейку и столбец с данными отделяет хотя бы одна пустая строка или вам нужен товар, который еще ни разу не вводился выше.

Способ 2. Стандартный

Этот способ чуть сложнее, но существенно богаче по возможностям настройки:

1. Выделите ячейки с данными, которые должны попасть в выпадающий список (например, наименования товаров).
2. Выделите ячейки (можно сразу несколько), в которых хотите получить выпадающий список, и выберите на вкладке **Данные – Проверка данных (Data – Validation)**. Из выпадающего списка **Тип данных (Allow)** выберите вариант **Список (List)**, поставьте курсор в поле **Источник (Source)** и выделите диапазон с эталонными значениями элементов списка:

	A	B	C	D	E	F	G	H
1	Наименование	Месяц	Склад	Продано				
2	Апельсины	Январь	#001	25			Апельсины	
3	Груши	Февраль	#001	311			Киви	
4	Киви	Апрель	#002	623			Яблоки	
5	Яблоки	Февраль	#002	207			Ананасы	
6	Апельсины	Март	#001	415			Бананы	
7							Сливы	
8								
9								
10								
11								
12								
13								
14								
15								
16								
17								
18								
19								
20								
21								
22								
23								
24								
25								
26								
27								
28								
29								
30								

Проверка вводимых значений ? [X]

Параметры Сообщение для ввода Сообщение об ошибке

Условие проверки

Тип данных:
 Список [v] Игнорировать пустые ячейки

Значение:
 между [v] Список допустимых значений

Источник:
 =\$G\$2:\$G\$7

Распространить изменения на другие ячейки с тем же условием


[Очистить все] [OK] [Отмена]

*Если диапазон источника расположен на другом листе, то в версиях Excel 2007 и старше его выделить не удастся. Нужно будет заранее дать ему имя (выделить и нажать **Ctrl+F3**) и ввести потом это имя в поле **Источник (Source)**. Не забудьте поставить перед именем знак "равно".*

Нажмите **OK** и проверьте результат:

	A	B	C
1	Наименование	Месяц	Склад
2	Апельсины	Январь	#001
3	Груши	Февраль	#001
4	Киви	Апрель	#002
5	Яблоки	Февраль	#002
6	Апельсины	Март	#001
7			
8	Апельсины		
9	Киви		
10	Яблоки		
11	Ананасы		
12	Бананы		
13	Сливы		
14			


Бонусом к этому способу идет возможность задать подсказку и сообщение об ошибке при неправильном вводе на вкладках **Сообщение для ввода (Input Message)** и **Сообщение об ошибке (Error Alert)**:



Способ 3. Элемент управления


Этот способ представляет собой вставку на лист нового объекта – элемента управления "Поле со списком" с последующей привязкой его к диапазонам на листе. Для этого:

1. Откройте вкладку **Разработчик (Developer)**. Если этой вкладки не видно, то в Excel 2007 нужно нажать кнопку **Офис – Параметры Excel** – флажок **Отображать вкладку Разработчик на ленте (Office Button – Excel Options – Show Developer Tab in the Ribbon)** или в Excel 2010–2013 щелкните правой кнопкой мыши по ленте, выберите команду **Настройка ленты (Customize Ribbon)** и включите отображение вкладки **Разработчик (Developer)** с помощью флажка.
2. Найдите значок выпадающего списка среди элементов управления форм (не ActiveX!). Ориентируйтесь по всплывающим подсказкам – **Поле со списком (Dropdown list)**:



Щелкните по значку и нарисуйте небольшой горизонтальный прямоугольник – будущий список. Чтобы рисовать ровно по границам ячеек – удерживайте клавишу Alt.

3. Щелкните по нарисованному списку правой кнопкой мыши и выберите команду **Формат объекта (Format control)**. В появившемся диалоговом окне задайте:



- **Формировать список по диапазону** – выделите ячейки с наименованиями товаров, которые должны попасть в список
- **Связь с ячейкой** – укажите ячейку, куда нужно выводить порядковый номер выбранного пользователем элемента.
- **Количество строк списка** – сколько строк показывать в выпадающем списке. По умолчанию – 8, но можно больше, чего не позволяет предыдущий способ.

После нажатия на **ОК** списком можно пользоваться.

Чтобы вместо порядкового номера элемента выводилось его название, можно дополнительно использовать функцию **ИНДЕКС (INDEX)**, которая умеет выводить содержимое нужной по счету ячейки из диапазона:


	A	B	C	D	E	F	G	H	I
1									
2		Клубника				3			
3						Клубника			
4									
5						Ананас			
6						Персик			
7						Клубника			
8						Малина			
9						Ежевика			
10						Черешня			
11						Яблоки			
12									
13									

Formula bar: `=ИНДЕКС(F5:F11;F2)`

Annotations: A red arrow points from the value '3' in cell F2 to the dropdown menu in cell B2. A red bracket on the right side of the table highlights the range F5:F11, labeled 'диапазон формирования' (formation range).


Способ 4. Элемент ActiveX

Этот способ частично напоминает предыдущий. Основное отличие в том, что на лист добавляется не элемент управления, а элемент ActiveX "Поле со списком" из раскрывающегося набора под кнопкой **Вставить (Insert)** с вкладки **Разработчик (Developer)**:




Механизм добавления тот же – выбираем объект из списка и рисуем его на листе. А вот дальше начинаются серьезные отличия от предыдущего способа.

Во-первых, созданный выпадающий ActiveX список может находиться в двух принципиально разных состояниях: режиме отладки, когда можно настраивать его параметры и свойства, двигать его по листу и менять размеры, и режиме ввода, когда единственное, что можно, – выбирать из него данные. Переключение между этими режимами происходит с помощью кнопки **Режим Конструктора (Design Mode)** на вкладке **Разработчик (Developer)**:



Если эта кнопка нажата, то мы можем настраивать параметры выпадающего списка, используя соседнюю кнопку **Свойства (Properties)**, которая откроет окно со списком всех возможных настроек для выделенного объекта, т.е. нашего списка:




Самые нужные и полезные свойства, которые можно и нужно настроить:

- **ListFillRange** – диапазон ячеек, откуда берутся данные для списка. Выделить мышью диапазон он не даст, надо просто вписать его руками с клавиатуры (например, Лист2!A1:A5);
- **LinkedCell** – связанная ячейка, куда будет выводиться выбранный из списка элемент;
- **ListRows** – количество отображаемых в выпадающем списке строк;
- **Font** – шрифт, размер, начертание (курсив, подчеркивание и т.д., кроме цвета);
- **ForeColor** и **BackColor** – цвет текста и фона соответственно.

Большим и жирным плюсом этого способа является возможность быстрого перехода к нужному элементу в списке при вводе первых букв с клавиатуры (!), чего нет у всех остальных способов.

Также возможно указывать в качестве **ListFillRange** не только одномерные диапазоны. Можно, например, задать диапазон из двух столбцов и нескольких строк, указав дополнительно, что выводить нужно два столбца (свойство **ColumnCount=2**). Тогда можно получить весьма привлекательные результаты, окупающие все потраченные на дополнительные настройки усилия:



Итоговая сравнительная таблица всех способов

	Способ 1. Примитивный	Способ 2. Стандартный	Способ 3. Элемент управления	Способ 4. Элемент ActiveX
Сложность	низкая	средняя	высокая	высокая
Возможность настройки шрифта, цвета и т.д.	нет	нет	нет	да
Количество отображаемых строк	всегда 8	всегда 8	любое	любое
Быстрый поиск элемента по первым буквам	нет	нет	нет	да
Необходимость использования дополнительной функции ИНДЕКС	нет	нет	да	нет
Возможность создания связанных выпадающих списков	нет	да	нет	нет

Связанные (зависимые) выпадающие списки

Под связанными списками понимаются несколько (минимум – два) выпадающих списков, когда содержимое последующих зависит от выбора пользователя в предыдущих. Например, в первом списке можно выбрать категорию товара, а во втором – увидеть товары из выбранной категории. Давайте рассмотрим несколько способов создать такие списки.

Способ 1. Функция ДВССЫЛ (INDIRECT)

Этот фокус основан на применении функции **ДВССЫЛ (INDIRECT)**, которая умеет делать одну простую вещь – преобразовывать содержимое любой указанной ячейки в адрес диапазона, который понимает Excel. Т.е. если в ячейке лежит текст "A1", то функция выдаст в результате ссылку на ячейку A1. Если в ячейке лежит слово "Маша", то функция выдаст ссылку на именованный диапазон с именем *Маша* и т.д.


Возьмем, к примеру, вот такой список моделей автомобилей Toyota, Ford и Nissan:

	A	B	C	D
1	Toyota	Ford	Nissan	
2	Altezza	Aspire	Almera	
3	Aristo	Bronco	Almera Tino	
4	Avalon	Capri	Altima	
5	Avensis	Contur	Armada	
6	Avensis Verso	Cougar	Auster	
7	Caldina	Crown Victoria	Avenir	
8	Cami	Econovan	Bassara	
9	Camry	Escape	Bluebird	
10	Camry Combi	Escort	Cedric	
11	Camry Gracia	Escort Cabrio	Largo	
12	Camry Prominent	Excursion	Laurel	
13	Carib	Expedition	Liberty	
14	Carina	Explorer	March	
15	Carina 2	F150	Maxima	
16	Carina E	Fiesta	Maxima QX	

Выделим весь список моделей Toyota (с ячейки A2 и вниз до конца списка) и дадим этому диапазону имя *Toyota* на вкладке **Формулы (Formulas)** с помощью **Диспетчера имен (Name Manager)**, кнопка **Создать (Create)**. Затем повторим то же самое со списками моделей Ford и Nissan, задав имена диапазонам *Ford* и *Nissan* соответственно.

При задании имен помните о том, что они не должны содержать пробелов, знаков препинания и начинаться обязательно с буквы. Поэтому если бы в одной из марок автомобилей присутствовал бы пробел (например, Land Rover), то его пришлось бы заменить в ячейке и в имени диапазона на нижнее подчеркивание (т.е. Land_Rover).

Теперь создадим первый выпадающий список для выбора марки автомобиля. Выделите пустую ячейку и нажмите кнопку **Проверка данных (Data Validation)** на вкладке **Данные (Data)**. Затем из выпадающего списка **Тип данных (Allow)** выберите вариант **Список (List)** и в поле **Источник (Source)** выделите ячейки с названиями марок (ячейки A1:C1 в нашем примере). После нажатия на **ОК** первый выпадающий список готов:



Теперь создадим второй (зависимый) выпадающий список, в котором будут отображаться только модели выбранной в первом списке марки. Так же как в предыдущем случае, откройте окно **Проверки данных**, но в поле **Источник** нужно будет ввести вот такую формулу:

=ДВССЫЛ(F3)

или =INDIRECT(F3)

где F3 – адрес ячейки с первым выпадающим списком (замените на свой).

Все. После нажатия на **ОК** содержимое второго списка будет выбираться по имени диапазона, выбранного в первом списке.

Минусы такого способа:

- В качестве вторичных (зависимых) диапазонов не могут выступать динамические диапазоны, задаваемые формулами типа **СМЕЩ (OFFSET)**. Для первичного (независимого) списка их использовать можно, а вот вторичный список должен быть определен жестко, без формул. Однако, это ограничение можно обойти, создав отсортированный список соответствий марка-модель (см. Способ 2).
- Имена вторичных диапазонов должны совпадать с элементами первичного выпадающего списка. Т.е. если в нем есть текст с пробелами, то придется их заменять на подчеркивания с помощью функции **ПОДСТАВИТЬ (SUBSTITUTE)**, т.е. формула будет выглядеть как =ДВССЫЛ(ПОДСТАВИТЬ(F3;" ";"_"))
- Надо руками создавать много именованных диапазонов (если у нас много марок автомобилей).

Способ 2. Список соответствий и функции **СМЕЩ (OFFSET)** и **ПОИСКПОЗ (MATCH)**

Этот способ требует наличия отсортированного списка соответствий марка-модель вот такого вида:

	A	B	C	D	E	F	G	H
1	Toyota	Altezza		Toyota				
2	Toyota	Aristo		Ford				
3	Toyota	Avalon		Nissan				
4	Toyota	Avensis						
5	Toyota	Avensis Verso						
6	Toyota	Caldina						
7	Toyota	Cami				Марка	Ford	
8	Ford	Aspire				Модель	Cougar	
9	Ford	Bronco					Aspire	
10	Ford	Capri					Bronco	
11	Ford	Contur					Capri	
12	Ford	Cougar					Contur	
13	Ford	Crown Victoria					Cougar	
14	Nissan	Almera					Crown Victoria	
15	Nissan	Almera Tino						

Для создания первичного выпадающего списка марок можно воспользоваться обычным способом, описанным выше, т.е.

- дать имя диапазону D1:D3 (например, *Марки*) с помощью **Диспетчера имен (Name Manager)** с вкладки **Формулы (Formulas)**
- выбрать на вкладке **Данные (Data)** команду **Проверка данных (Data Validation)**
- выбрать из выпадающего списка вариант проверки **Список (List)** и указать в качестве **Источника (Source)** =*Марки* или просто выделить ячейки D1:D3 (если они на том же листе, где список).

А вот для зависимого списка моделей придется создать именованный диапазон с функцией **СМЕЩ (OFFSET)**, который будет динамически ссылаться только на ячейки моделей определенной марки. Для этого:

- Нажмите **Ctrl+F3** или воспользуйтесь кнопкой **Диспетчер имен (Name Manager)** на вкладке **Формулы (Formulas)**.
- Создайте новый именованный диапазон с любым именем (например, *Модели*) и в поле **Ссылка (Reference)** в нижней части окна введите руками следующую формулу:

```
=СМЕЩ($A$1;ПОИСКПОЗ($G$7;$A:$A;0)-1;1;СЧЁТЕСЛИ($A:$A;$G$7);1)
```


```
=OFFSET($A$1;MATCH($G$7;$A:$A;0)-1;1;COUNTIF($A:$A;$G$7);1)
```

Ссылки должны быть абсолютными (со знаками \$). После нажатия **Enter** к формуле будут автоматически добавлены имена листов – не пугайтесь.

Функция **СМЕЩ (OFFSET)** умеет выдавать ссылку на диапазон нужного размера, сдвинутый относительно исходной ячейки на заданное количество строк и столбцов. В более понятном варианте синтаксис этой функции таков:

```
=СМЕЩ(начальная_ячейка; сдвиг_вниз; сдвиг_вправо; размер_диапазона_в_строках;
размер_диапазона_в_столбцах)
```

```
=СМЕЩ(старт; A; B; C; D)
```




Таким образом:

- начальная ячейка – берем первую ячейку нашего списка, т.е. A1
- сдвиг_вниз – считает функция **ПОИСКПОЗ (MATCH)**, которая, попросту говоря, выдает порядковый номер ячейки с выбранной маркой (G7) в заданном диапазоне (столбце A)
- сдвиг_вправо = 1, т.к. мы хотим сослаться на модели в соседнем столбце (B)
- размер_диапазона_в_строках – вычисляем с помощью функции **СЧЁТЕСЛИ (COUNTIF)**, которая умеет подсчитать количество встретившихся в списке (столбце A) нужных нам значений – марок авто (G7)
- размер_диапазона_в_столбцах = 1, т.к. нам нужен один столбец с моделями

В итоге должно получиться что-то вроде этого:

	A	B	C	D	E	F	G	H	I	J
1	Toyota	Altezza		Toyota						
2	Toyota	Aristo		Ford						
3	Toyota	Avalon		Nissan						
4	Toyota	Avensis								
5	Toyota	Avensis Verso								
6	Toyota	Caldina								
7	Toyota	Cami				Марка	Ford			
8	Ford	Aspire				Модель	Cougar			
9	Ford	Bronco								
10	Ford	Capri								
11	Ford	Contur								
12	Ford	Cougar								
13	Ford	Crown Victoria								
14	Nissan	Almera								
15	Nissan	Almera Tino								
16	Nissan	Altima								
17	Nissan	Armada								
18	Nissan	Auster								
19	Nissan	Avenir								
20	Nissan	Bassara								
21	Nissan	Bluebird								
22	Nissan	Cedric								
23										
24										
25										
26										
27										
28										
29										
30										
31										
32										



Осталось добавить выпадающий список на основе созданной формулы к ячейке G8. Для этого:

- выделяем ячейку G8
- выбираем на вкладке **Данные (Data)** команду **Проверка данных (Data Validation)**
- из выпадающего списка выбираем вариант проверки **Список (List)** и вводим в качестве **Источника (Source)** знак "равно" и имя нашего диапазона, т.е. *=Модели*

Выпадающий список для выбора изображений товаров

Необходимо сделать так, чтобы в одной из ячеек листа был выпадающий список с наименованиями, при выборе из которого товар отображался бы рядом в виде фотографии:

	A	B	C
1	Nokia 8800		
2			
3			
4			

Шаг 1. Создаем каталог с фото и даем ему имя


Создаем на **Листе 1** каталог с наименованиями и фотографиями товаров, состоящий из двух столбцов (**Модель** и **Фото**):

	A	B	C
1	Модель	Фото	
2	Nokia 8800		
3	Voxtel W420		
4	Motorola RAZR V3i		
5	SonyEricsson M600i		
6			

Теперь надо дать имя нашему каталогу, чтобы ссылаться на него в будущем. Для этого используем кнопку **Диспетчер имен (Name Manager)** на вкладке **Формулы (Formulas)**. Создаем новый диапазон, вводим имя (например, *Фотоальбом*) и в качестве адреса указываем формулу:

```
=СМЕЩ(Лист1!$A$1;1;0;СЧЁТЗ(Лист1!$A:$A)-1;1)
```


```
=OFFSET(Лист1!$A$1;1;0;COUNTA(Лист1!$A:$A)-1;1)
```



Эта формула определяет последнюю занятую ячейку в столбце А и выдает на выходе диапазон с А2 до этой найденной ячейки. Такая относительно сложная конструкция нужна, чтобы впоследствии дописывать новые модели к нашему списку и не думать об исправлении диапазона. Если дописывать точно ничего не придется, то можете вместо ввода этой страшноватой формулы просто указать =А2:А5

Шаг 2. Выпадающий список для выбора модели


Перейдем на **Лист 2** и создадим там ячейку с выпадающим списком для выбора пользователем модели телефона (пусть это будет А1). Выделяем ячейку и идем на вкладку **Данные – Проверка данных (Data – Data Validation)**. Далее в поле **Тип данных (Allow)** выбираем **Список (List)**, а в качестве **Источника (Source)** указываем наш *Фотоальбом* (не забудьте перед ним добавить знак равенства):




Кроме того, этой ячейке удобно дать имя – вкладка **Формулы – Диспетчер имен (Formulas – Name Manager)**, создаем диапазон и далее вводим имя (например, *Выбор*) и **ОК**.

Шаг 3. Копируем фотографию

Перенесем первую фотографию из фотоальбома к выпадающему списку. Надо выделить ячейку с первой фотографией (не сам рисунок, а ячейку!) и развернуть выпадающий список под кнопкой **Копировать (Copy)** на **Главной (Home)** вкладке:



Microsoft Excel 2007 больше ничего не спросит, а в Excel 2010 появится еще одно дополнительное окно с выбором типа создаваемого изображения:




В нем нужно выбрать варианты "как на экране" и "растровый".

Копируем, переходим на **Лист 2** к выпадающему списку и в любую пустую ячейку недалеко от него вставляем наш мини-скриншот ячейки с фотографией (на вкладке **Главная – Правка – Вставить** или обычное **CTRL+V**).

Шаг 4. Создаем динамическую ссылку на выбранную фотографию

Теперь необходимо сделать ссылку, которая будет указывать на ячейку с выбранной фотографией. Снова открываем **Диспетчер имен (Name Manager)** на вкладке **Формулы (Formulas)** и создаем еще один именованный диапазон:



...или в английском варианте

=OFFSET(Лист1!\$B\$2;MATCH(Выбор;Фотоальбом;0)-1;0;1;1)


Технически функция **ПОИСКПОЗ (MATCH)** находит ячейку с нужной моделью в каталоге по названию, а функция **СМЕЩ (OFFSET)** затем выдает ссылку на соседнюю справа от найденного названия ячейку, т.е. ячейку с фотографией товара.

Шаг 5. Привязываем фотографию к ссылке

Осталось выделить скопированную фотографию на **Листе 2** и вписать в строку формул

=Фото

и нажать Enter



Выпадающий список с автоматическим добавлением отсутствующих элементов

Задача: сделать в ячейке D2 выпадающий список, чтобы пользователь мог выбирать имена из списка (столбец А). Если нужного имени нет в списке, то пользователь может ввести новое имя прямо в ячейку D2 – оно автоматически добавится к столбцу А и начнет отображаться в выпадающем списке в будущем.

	A	B	C	D	E
1	Саша				
2	Маша		Имя сотрудника:	<input type="text"/>	
3	Петя				
4	Вася				
5	Коля				
6	Лена				
7	Иван				
8	Миша				
9	Махмуд				
10	Ирина				
11					
12					

Шаг 1. Создаем именованный диапазон

Сначала создадим именованный диапазон, указывающий на заполненные именами ячейки в столбце А – сколько бы имен в списке ни находилось. Для этого жмем на вкладке **Формулы (Formulas)** кнопку **Диспетчер имен (Name Manager)** и затем **Создать (New)**.

Вводим имя диапазона (допустим, *People*) и в строку **Ссылка (Reference)** вводим следующую формулу:

```
=СМЕЩ(Лист1!$A$1;0;0;СЧЁТЗ(Лист1!$A$1:$A$24);1)
```


в английской версии Excel это будет:

```
=OFFSET(Лист1!$A$1;0;0;COUNTA(Лист1!$A$1:$A$24);1)
```

Эта формула ссылается на все заполненные ячейки в столбце А, начиная с А1 и вниз до конца – до последнего имени.

Шаг 2. Создаем выпадающий список в ячейке

Выделяем ячейку D2 и жмем на вкладке **Данные (Data)** кнопку **Проверка данных (Data Validation)**. Далее выбираем из выпадающего списка **Тип данных (Allow)** позицию **Список (List)** и вводим в строку **Источник (Source)** ссылку на созданный на шаге 1 именованный диапазон (не забудьте перед именем диапазона поставить знак равенства!):



Чтобы Excel позволил нам в будущем ввести в список и новые имена, снимем галочки на вкладках **Сообщение для ввода (Input Message)** и **Сообщение об ошибке (Error Alert)** и нажмем **ОК**. Теперь у нас есть выпадающий список в ячейке D2. Причем если, например, вручную дописать новое имя в столбце A, то оно автоматически появится в выпадающем списке в ячейке D2, поскольку имена берутся из динамического диапазона *People*, который автоматически отслеживает изменения в столбце A.

Шаг 3. Добавляем простой макрос


Щелкаем правой кнопкой мыши по ярлычку нашего листа и выбираем **Исходный текст (View Source)**. Откроется модуль листа в редакторе Visual Basic, куда надо ввести такой код:

```
Private Sub Worksheet_Change(ByVal Target As Range)
Dim lReply As Long

If Target.Cells.Count > 1 Then Exit Sub
If Target.Address = "$D$2" Then
If IsEmpty(Target) Then Exit Sub
If WorksheetFunction.CountIf(Range("People"), Target) = 0 Then
lReply = MsgBox("Добавить введенное имя " & _
Target & " в выпадающий список?", vbYesNo + vbQuestion)
If lReply = vbYes Then
Range("People").Cells(Range("People").Rows.Count + 1, 1) = Target
End If
End If
End If
End Sub
```

Если ваш выпадающий список находится не в ячейке D2 или вы назвали диапазон с именами не *People*, а как-то еще, то подправьте эти параметры в макросе на свои.

Всё! Теперь при попытке ввести новое имя в ячейку D2 Excel будет спрашивать



...и при утвердительном ответе пользователя автоматически добавлять новое имя к списку в столбце А и в выпадающий список в ячейку D2.

Выпадающий список с удалением использованных элементов

Постановка задачи

Имеем в качестве примера недельный график дежурств, который надо заполнить именами сотрудников, причем для каждого сотрудника максимальное количество рабочих дней (смен) ограничено. Идеальным вариантом было бы организовать в ячейках B2:B8 выпадающий список, но при этом сделать так, чтобы уже занятые сотрудники автоматически убрались из выпадающего списка, оставляя только свободных:

	A	B	C	D	E	F	G	H	I
1	День недели	Дежурный							
2	Понедельник			№	Сотрудники	Допустимо смен	Смен назначено	Номера свободных сотрудников	Неиспользованные сотрудники
3	Вторник		1	Александр	1	0	1	Александр	
4	Среда		2	Мария	1	0	2	Мария	
5	Четверг		2	Елена	2	0	3	Елена	
6	Пятница		1	Николай	1	0	4	Николай	
7	Суббота		1	Петр	1	0	5	Петр	
8	Воскресенье		0	Василиса	0	0		Егор	
9			1	Егор	1	0	7	Кирилл	
10			3	Кирилл	3	0	8	Лариса	
11			1	Лариса	1	0	9		
12									
13									

Чтобы реализовать подобный вариант выпадающего списка, выполним несколько простых шагов.

Шаг 1. Кто сколько работает?

Сначала давайте подсчитаем, кто из наших сотрудников уже назначен на дежурство и на сколько смен. Для этого добавим к зеленой таблице еще один столбец, введем в него следующую формулу:

=СЧЁТЕСЛИ(\$B\$2:\$B\$8;E2) или в англоязычной версии =COUNTIF(\$B\$2:\$B\$8;E2)

G2		fx		=СЧЁТЕСЛИ(\$B\$2:\$B\$8;E2)			
	A	B	C	D	E	F	G
1	День недели	Дежурный					
2	Понедельник	Александр		№	Сотрудники	Допустимо смен	Смен назначено
3	Вторник		1	1	Александр	1	1
4	Среда			2	Мария	1	0
5	Четверг	Кирилл		2	Елена	2	0
6	Пятница	Николай		1	Николай	1	1
7	Суббота			1	Петр	1	0
8	Воскресенье			0	Василиса	0	0
9				1	Егор	1	0
10				3	Кирилл	3	1
11				1	Лариса	1	0

Фактически формула просто вычисляет, сколько раз имя сотрудника встречалось в диапазоне с именами.

Шаг 2. Кто еще свободен?

Теперь выясним, кто из наших сотрудников еще свободен, т.е. не исчерпал запас допустимых смен. Добавим еще один столбец и введем в него формулу, которая будет выводить номера свободных сотрудников:

=ЕСЛИ(F2-G2<=0;"";СТРОКА(E2)-СТРОКА(\$E\$2)+1)

или в англоязычной версии =IF(F2-G2<=0;"";ROW(E2)-ROW(\$E\$2)+1)

H2		fx =ЕСЛИ(F2-G2<=0;"";СТРОКА(E2)-СТРОКА(\$E\$2)+1)						
	A	B	C	D	E	F	G	H
	День недели	Дежурный						
1								
2	Понедельник	Александр						
3	Вторник							
4	Среда							
5	Четверг	Кирилл						
6	Пятница	Николай						
7	Суббота							
8	Воскресенье							
9								
10								
11								

№	Сотрудники	Допустимо смен	Смен назначено	Номера свободных сотрудников
1	Александр	1	1	
2	Мария	1	0	2
3	Елена	2	0	3
4	Николай	1	1	
5	Петр	1	0	5
6	Василиса	0	0	
7	Егор	1	0	7
8	Кирилл	3	1	8
9	Лариса	1	0	9

Шаг 3. Формируем список

Теперь надо сформировать непрерывный (без пустых ячеек) список свободных сотрудников для связи – на следующем шаге – с выпадающим списком. Для этого добавим еще один столбец и введем в него такую страшноватую на первый взгляд формулу:

=ЕСЛИ(D2>СЧЁТ(\$H\$2:\$H\$10);"";ИНДЕКС(\$E\$2:\$E\$10;НАИМЕНЬШИЙ(\$H\$2:\$H\$10;СТРОКА(E2)-1)))

или соответственно

=IF(D2>COUNT(\$H\$2:\$H\$10);"";INDEX(\$E\$2:\$E\$10;SMALL(\$H\$2:\$H\$10;ROW(E2)-1)))

I2		fx =ЕСЛИ(D2>СЧЁТ(\$H\$2:\$H\$10);"";ИНДЕКС(\$E\$2:\$E\$10;НАИМЕНЬШИЙ(\$H\$2:\$H\$10;СТРОКА(E2)-1)))								
	A	B	C	D	E	F	G	H	I	J
	День недели	Дежурный								
1										
2	Понедельник	Александр							Мария	
3	Вторник								Елена	
4	Среда								Петр	
5	Четверг	Кирилл							Егор	
6	Пятница	Николай							Кирилл	
7	Суббота								Лариса	
8	Воскресенье									
9										
10										
11										

№	Сотрудники	Допустимо смен	Смен назначено	Номера свободных сотрудников	Неиспользованные сотрудники
1	Александр	1	1		
2	Мария	1	0	2	
3	Елена	2	0	3	
4	Николай	1	1		
5	Петр	1	0	5	
6	Василиса	0	0		
7	Егор	1	0	7	
8	Кирилл	3	1	8	
9	Лариса	1	0	9	

При всей внешней жуткости вида эта формула делает одну простую вещь – выводит очередное по номеру имя сотрудника (используя функцию НАИМЕНЬШИЙ) из списка или пустую ячейку, если имена свободных сотрудников уже кончились.


Шаг 4. Создаем именованный диапазон свободных сотрудников

Жмем кнопку **Диспетчер имен (Name Manager)** на вкладке **Формулы (Formulas)** и создаем новый именованный диапазон *Имена* по следующей формуле:

=СМЕЩ(Лист1!\$I\$2;0;0;СЧЁТ3(Лист1!\$I\$2:\$I\$10)-СЧИТАТЬПУСТОТЫ(Лист1!\$I\$2:\$I\$10))

...или в англоязычной версии:


=OFFSET(Лист1!\$I\$2;0;0;COUNTA(Лист1!\$I\$2:\$I\$10)-COUNTBLANK(Лист1!\$I\$2:\$I\$10))



Фактически мы просто даем диапазону занятых ячеек в синем столбце собственное название *Имена*.

Шаг 5. Создаем выпадающий список в ячейках

Осталось выделить ячейки B2:B8 нашего графика и добавить в них выпадающий список с элементами диапазона *Имена*. Для этого жмем кнопку **Проверка данных (Data Validation)** на вкладке **Данные (Data)** и в открывшемся окне выберем в списке допустимых значений вариант **Список (List)** и укажем **Источник (Source)** данных:



Вот и все! Теперь при назначении сотрудников на дежурство их имена будут автоматически удаляться из выпадающего списка, оставляя только тех, кто еще свободен.


Выпадающий список на основе динамического диапазона

Задача: создать в ячейке выпадающий список для удобного ввода информации. Варианты для списка должны браться из заданного динамического диапазона, т.е. если завтра в него внесут изменения – например, удалят ненужные элементы или допишут еще несколько новых, – они должны автоматически отразиться в выпадающем списке:

	A	B	C	D	E
1	<i>Сотрудники</i>				
2	Иван				
3	Мария				
4	Дарья				
5	Василий				
6					
7					


Для реализации такой задачи есть простой и удобный способ почти без формул. Способ основан на новой возможности последних версий Microsoft Excel начиная с 2007 версии. Суть его в том, что любой диапазон можно выделить и отформатировать как Таблицу. Тогда он превращается, упрощенно говоря, в "резиновый", т.е. сам начинает отслеживать изменения своих размеров, автоматически растягиваясь-сжимаясь при добавлении-удалении в него данных.

Выделите диапазон вариантов для выпадающего списка (A1:A5 в нашем примере выше) и на **Главной (Home)** вкладке нажмите кнопку **Форматировать как таблицу (Home – Format as Table)**. Дизайн можно выбрать любой – это роли не играет:




Обратите внимание на то, что таблица должна иметь строку заголовка (в нашем случае это A1 со словом *Сотрудники*). Первая ячейка играет роль "шапки" и содержит название столбца. На появившейся после

превращения в Таблицу вкладке **Конструктор (Design)** можно изменить стандартное имя таблицы на свое (без пробелов!). По этому имени мы сможем потом адресоваться к таблице на любом листе этой книги:



Теперь выделите ячейки, где вы хотите создать выпадающие списки (в нашем примере выше это D2), и нажмите кнопку **Проверка данных (Data Validation)** на вкладке **Данные (Data)**. В открывшемся окне на вкладке **Параметры (Settings)** выберите вариант **Список (List)** и введите в поле **Источник (Source)** вот такую формулу:




=INDIRECT("Таблица1[Сотрудники]")


Смысл этой формулы прост. Выражение *Таблица1[Сотрудники]* – это ссылка на столбец с данными для списка из нашей умной таблицы. Но проблема в том, что Excel почему-то не хочет понимать прямых ссылок в поле **Источник (Source)**, т.е. нельзя написать в поле **Источник** выражение вида =Таблица1[Сотрудники]. Поэтому мы идем на тактическую хитрость – вводим ссылку как текст (в кавычках) и используем функцию **ДВССЫЛ (INDIRECT)**, которая преобразовывает текстовую ссылку в настоящую, "живую".

Осталось только нажать на **OK**. Если теперь дописать к нашей таблице новые элементы, то они будут автоматически в нее включены, а значит – добавятся к нашему выпадающему списку. С удалением – то же самое.

Если вам лень возиться с вводом формулы ДВССЫЛ, то можно чуть упростить процесс. После создания умной таблицы просто выделите мышью диапазон с элементами для выпадающего списка (A2:A5) и введите в поле адреса имя для этого диапазона (без пробелов), например *Стажеры*, и нажмите на **Enter**:



Фактически этим мы создаем именованный динамический диапазон, который ссылается на данные из нашей умной таблицы. Теперь имя этого диапазона можно ввести в окне создания выпадающего списка в поле **Источник (Source)**:



Выпадающий список с данными из другого файла


Создать простой выпадающий список в ячейке с данными из этого же файла – просто. Все усложняется, когда сам список и данные для него находятся в разных файлах. Для решения этой проблемы есть несколько способов.

Способ 1. Функция ДВССЫЛ

В простом случае можно использовать функцию **ДВССЫЛ (INDIRECT)**, чтобы сформировать правильную ссылку на внешний файл.

Например, если необходимо создать выпадающий список с содержимым ячеек A1:A10 с листа *Список* из файла *Товары.xls*, нужно открыть окно проверки данных через вкладку **Данные – Проверка данных (Data – Validation)** и в поле **Источник (Source)** ввести следующую конструкцию:

=ДВССЫЛ("[Товары.xls]Список!\$A\$1:\$A\$10")



Функция **ДВССЫЛ (INDIRECT)** преобразует текстовую строку аргумента в реальный адрес, используемый для ссылки на данные. Обратите внимание, что имя файла заключается в квадратные скобки, а восклицательный знак служит разделителем имени листа и адреса диапазона ячеек. Если имя файла содержит пробелы, то его надо заключить в апострофы.

Если файл с исходными данными для списка лежит в другой папке, необходимо указать полный путь к файлу, например, следующим образом:

=ДВССЫЛ("'D:\Договоры\[Товары.xls]Список!\$A\$1:\$A\$10")

В данном случае не забудьте заключить в апострофы полный путь к файлу и имя листа.

Минус этого способа только один – выпадающий список будет корректно работать только в том случае, если файл *Товары.xls* открыт.


Способ 2. Импорт данных

Суть этого способа в том, что данные для выпадающего списка будут импортироваться из другого файла в текущий. Для этого необходимо создать обновляемую связь между двумя файлами.


Сначала откройте файл-источник, где находятся эталонные значения для выпадающего списка (назовем его, допустим, **Справочник.xlsx**).

Выделите диапазон с данными для списка и отформатируйте его как таблицу с помощью кнопки **Форматировать как таблицу** на вкладке **Главная (Home – Format as Table)**. Обратите внимание, что у такой таблицы предварительно должна быть сделана "шапка" – строка заголовка. После этого файл *Справочник* можно сохранить и закрыть.


Теперь откроем книгу, где мы хотим создать выпадающий список (условно назовем ее **Бланк.xlsx**). Вставим чистый лист (**Alt+F11**), выберем на вкладке **Данные – Существующие подключения – Найти другие (Data – Existing Connections – Browse for more)** и укажем наш файл **Справочник.xlsx**. Появится диалоговое окно, в котором Excel спросит нас о том, какую именно таблицу мы хотим импортировать (если их в файле было несколько):



После нажатия на **OK** появится еще одно последнее окно, где можно указать удобную ячейку для импорта и, нажав на кнопку **Свойства (Properties)**, задать частоту обновления информации:




Тут можно включить флажок **Обновить при открытии файла (Refresh on open)**, чтобы каждый раз при открытии этой книги иметь последнюю версию списка:



После нажатия на кнопку **OK** Excel загрузит данные из созданной таблицы из файла *Справочник* в наш текущий файл *Бланк* и отформатирует их в виде таблицы:

	A	B
1	Товары ▾	
2	Ананасы	
3	Брюква	
4	Груша	
5	Салат	
6	Огурцы	
7	Яблоки	
8		


Если выделить импортированный список (диапазон A2:A7 в нашем случае), то в строке формул можно увидеть его имя, которое он автоматически получает при вставке:



Это имя также можно увидеть в **Диспетчере имен** на вкладке **Формулы (Formulas – Name Manager)**.

Осталось создать выпадающий список, который будет ссылаться на эти данные. Для этого:

1. Выделяем ячейки, где хотим создать выпадающие списки.
2. На вкладке **Данные** жмем на кнопку **Проверка данных (Data – Validation)**.
3. Выбираем в раскрывающемся списке разрешенных типов данных вариант **Список (List)** и вводим в поле **Источник (Source)** следующую формулу:



В англоязычной версии Excel это будет =INDIRECT("Таблица_Справочник").

Логично было бы ввести просто имя нашего диапазона, но, к сожалению, Microsoft Excel почему-то не воспринимает имена таблиц в поле **Источник**. Поэтому мы используем тактическую хитрость – функцию **ДВССЫЛ (INDIRECT)**, которая превращает свой аргумент (имя нашей таблицы) в рабочую ссылку.

Все. После нажатия на **ОК** наш список начнет работать и будет в дальнейшем автоматически обновляться, подгружая данные из другого файла:

